Newsgroup: it.comp.software.newsreader
Titolo: [Hamster] Convertire una mailing-list in un newsgroup locale

[spero che questo sia il ng giusto]
Premesso che non so utilizzare gli script di Hamster (ma questa è
un'ottima occasione per imparare ^___^;;;) pongo il seguente quesito:
È possibile ricevere i messaggi da una mailing-list (nel mio caso una
mailing list di Yahoo e-groups) dal solito account di posta e
convertirli tramite uno script in modo che Free Agent li possa vedere
come un newsgroup sul server locale?
Qualcuno lo ha già fatto (e mi può passare lo script, magari...)?
Mi interessa in particolare ottenere la ramificazione delle risposte
(come fa OE) e posso tranquillamente soprassedere sui problemi di
posting (tanto lurko e basta :)
Grazie in anticipo
Ciao
-- Gov

 Qualcuno lo ha già fatto (e mi può passare lo script, magari...)?
e' quello che faccio io con quasi tutte le mie ML e non occorrono script
basta che ti crei dei newsgroup locali (meglio se li chiami 
internal.mailinglist.*)
e poi nelle regole dei messaggi inserisci:
#mailing list
postto(internal.mailinglist.cristinadona) 	Any-Recipient: 
"
cristi...@listbot.com
"
ovviamente ricordati che per rispondere non devi fare un follow-up ma un 
reply-to... :)

 come un newsgroup sul server locale?
qche giorno fa ci fu un thread apposta (*[hamster] posta con gravity*, o 
qsa del genere)
ad ogni modo crea il tuo ng locale
aggiungi nel filtro della posta
[*]
postto(nomedelnglocale) Any-Recipient: "indirizzodellaml"
ne' piu' ne' meno :)
-- 
Josef K.
jos...@freemail.it

On Mon, 16 Jul 2001 12:05:15 GMT, 
governor_...@libero.it
(Governor LePetomane) wrote:
ok... fatto.
Ora, già che ci sono (grazie a tutti, comunque) è possibile portare
sul newsgroup locale anche l'archivio dei messaggi che ho finora
ricevuto sotto OE?
Ho già il DbXtract.exe che mi dà in uscita dei .eml, mi manca il
passaggio
                .eml => .msg
e il passaggio
                .msg => newsgroup locale.
soprattutto perché non riesco a far applicare le regole a messaggi che
non provengano da POP3, nè tantomeno riesco a inoltrare tali messaggi
direttamente sul newsgroup.
forse c'è un modo di creare un POP3 "locale" che dia in pasto ad
Hamster e quindi alle sue regole dei messaggi già presenti sull'HD?
questa sarebbe una soluzione (suppongo)...
-- Gov

 
Se trasformi l'archivio in formato unix puoi usare questo script che ho 
fatto tempo fa:
Nelle prime righe devi mettere i tuoi dati 
#!hs2
var($filembx,$Linee,$LineeTemp,$i,$linea,$primicar)
var($nomeGruppo,$trovato)
#inserire in $nomeGruppo il nome del gruppo locale
$nomeGruppo="mailing-list.prova"
#inserire in $filembx il nome del file mbx completo del percorso
$filembx="E:\PMAIL\MAIL\Sandro\UNX02501.MBX"
hamwaitidle
if (!fileExists($filembx))
   warning ("Il file " + $filembx + " non esiste")
   quit
endif
$trovato=0
$i = 0
while ($i < HamGroupCount)
    if (HamGroupName($i)=$nomeGruppo)
	$trovato=1
	break
    endif
    inc ($i)
endwhile
if (!$trovato)
   warning ("Il gruppo " + $nomeGruppo + " non esiste")
   quit
endif
$Linee = listalloc()
$LineeTemp = listalloc()
ErrCatch (True)
listload($Linee,$filembx)
if (ErrNum<>0)
    if (ErrNum=3)
        warning("File " + $filembx + " probabilmente gia' in uso"
    else
        warning ("Errore: ", ErrNum, " ", ErrMsg)
    endif
    listfree($Linee )
    listfree($LineeTemp )
    quit
endif
ErrCatch (false)
for ($i,0,listcount($Linee)-1)
   $linea= listget($Linee,$i)
   listadd($LineeTemp,$linea)
   $primicar = copy($linea, 1, 5)
   if ($primicar="From:")
   	listadd($LineeTemp,"Newsgroups: " + $nomeGruppo)	
   endif
endfor
ListSave ($LineeTemp, "tempmbx.txt")
execute ("ham import -ih -go""" + $nomeGruppo + """ " + _
"tempmbx.txt")
FileDelete ("tempmbx.txt")
listfree($Linee )
listfree($LineeTemp )
quit